
Keypad Display Wildcard

The Keypad/Display WildcardTM provides a convenient interface to a 4 x 20 character display
and 4 x 5 keypad. Combined with Mosaic's QCard, QScreen or QVGA Controllers, it is an
ideal solution for hand-held or space-constrained applications that require a programmable
embedded computer and a low-cost yet smart user interface.

Measuring only 2” x 2.5”, the Keypad/Display Wildcard mounts directly on a QCard
Controller.

This Wildcard is shipped with a 4x20 LED-backlit character display and 4x5 keypad,
plus a simple ribbon cable interface for custom placement of the keypad and display in
your instrument.

It has an additional field header that brings out 4 nibble-wise programmable input/output
lines and 4 input lines.

This WildCard also contains an onboard 2 KHz buzzer to provide audible feedback for
keypad presses, or any other purpose.

The Keypad/Display WildCard provides a hardware and software interface for a 5x4 keypad and
4x20 liquid crystal display (LCD).  These devices connect to the WildCard via a simple “straight-
through” ribbon cable interface.  Pre-coded routines in the QED-Forth kernel (available to both C
and Forth programmers) scan the keypad and write to the LCD display.

Table 1-1 Technical Specifications

Property Value

Power 5 VDC derived from the WildCard bus
0.5 W using the non-backlit display), or,
2 W using an LED-backlit display.

Keypad 5 column by 4 row keypad, tactile feedback, snap-on domes for user-configurable
legends, mounting hole size 2.7” x 3.0”, standard Grayhill part, interchangeable
with other sizes.

Display 4 line by 20 character LCD display with optional LED backlight

Backlight LED backlight with software ON/OFF.

Beeper Software controlled 2 KHz, 0.2 W buzzer at 80 dB

General Purpose I/O 4 input lines and 4 lines programmable together as all inputs or all outputs

Output current capability: 4mA source, 24mA sink

Property Value
Connectors 34-pin dual row 0.1" pitch keypad/display connector

24-pin dual row 0.1" pitch field I/O connector

Connecting the WildCard
The WildCard is shown in Figure 1.1.  On the right the WildCard Port Header, H1, connects to the
QCard or QScreen Controller, and on the left the Field Header, H4, provides 8 lines of digital I/O.
The 34-pin connector on the side, H2, connects to a ribbon cable to the keypad and display.

H2: Keypad and Display Header

Figure 1-1 Headers and Jumpers on the Keypad/Display WildCard.

Mounting on a QCard, QScreen, or QVGA Controller

With the power off, the WildCard may be mounted on a QCard, QScreen, QVGA Controller, or
PowerDock by directly plugging connector H1 into a WildCard Port connector on the controller.
The corner mounting holes on the module should line up with the standoffs on the controller.

Setting the Module Address

Each WildCard Port on a QCard, QScreen or QVGA Controller accommodates up to four
WildCards, at four different module addresses: 0, 1,2, or 3 if installed on Port 0, and 4, 5, 6, or 7 if
installed on Port 1.  A QCard has only one module port, Port 0.  Even so, if the QCard is mounted

on a PowerDock it can hold four WildCards on the QCard, and four more on Port 1 of the Power-
Dock.  You should set the module addresses on the WildCards in your system so that they do not
conflict, that is, so that no two modules on the same port have the same address (jumper settings).
Jumpers J1 and J2 allow you to select the module address.

In order to use the built-in driver routines the Keypad/Display WildCard must be
configured at module address 0.  In some applications you may wish to forgo use of
the built-in drivers and configure the WildCard at another module address.  In that
case, use Table 1-2 to determine the jumper settings you need for each module ad-
dress.   But to make use of the internal driver routines, you must configure the
WildCard for module address 0.  Make sure that there are no jumpers set on J1 and
J2.

Table 1-2 Jumper settings and addresses.

Module Port Module Address Page Address Installed Jumper Shunts

0 0x00 None

1 0x01 J1

2 0x02 J2
0

3 0x03 J1 and J2

4 0x04 None

5 0x05 J1

6 0x06 J2
1

7 0x07 J1 and J2

Setting the Jumpers

There are six jumpers on the board allowing you to select various options.  Table xx describes their
purposes and default settings.

Table 1-3 Default jumper positions.

Jumper Default Setting Meaning

J1 Removed Module address bit.  Together with J2 chooses module addresses of 0-3.
Should be removed in normal operation to choose page 0 addresses.

J2 Removed Module address bit.  Should be removed in normal operation to choose
page 0 addresses.

J3 Connected Enables the onboard beeper.  Remove to disable the beeper.

J4 Connected Enables the onboard contrast adjustment potentiometer.  Remove if an
external pot is used to adjust the display contrast.

J5 Connected Chooses the display type: connected for an Hitachi character or graphics
display controller; or removed for a Toshiba graphics display controller.
Standard WildCards ship with an Hitachi-type character display.

Jumper Default Setting Meaning
J6 Set to “Char” Chooses either a 4x20 character or a graphics display.  Standard

WildCards ship with a character display.

Connecting a Keypad and Display

The keypad/display interface connector, H2 in the photo, is a 34-pin dual-row right-angled header.
A 34-line ribbon cable plugs into this header.  At the other end of the cable are two female single-
row connectors: one at the end for the keypad, and the other in the middle for the display.

The 34-pin female connector should be plugged into H2 with the ribbon exiting the connector to-
ward the board.  Line 1 in the ribbon cable (on the edge of the cable that the 10-pin keypad connec-
tor is aligned along) mates with pin 1 of H2, closest to the field header, H4.

Connecting the Keypad

The 10-pin connector on the end of the ribbon cable should connect to the keypad.  Take care – the
keypad has only 9 pins!  The connector should be attached so that pin 1 on the keypad (designated
by the “F” on the Grayhill 20-key keypad, Part No. 86JB2), and on the left looking at the back of the
keypad with the connector at the top), corresponds to pin 1 on the cable (the edge of the cable that
terminates on the WildCard closest to the field header, H4).  In operation, the keypad is oriented so
that its connector is on the back at the bottom of the keypad.

Connecting the Display

The WildCard can accommodate liquid crystal display modules up to 4 lines by 20 characters in
size.  If your display still has a clear plastic cover to protect the display glass from scratches during
shipment don’t forget to peel it off.

The processor sends control commands and ASCII data to the display and powers its backlight
through a 16-pin bus.  The 16-pin single-row female connector on the ribbon cable connects directly
to the display pins at the top rear of the display.  The edge of the cable to connect to pin 1 of the
display (nearest the display’s corner) is the one aligned with pin 1 of the WildCard’s header H2, and
closest to the field header on the WildCard.  It is also the edge closest to the keypad connector.  In
its standard orientation, the display connector is at the top back of the display, and the display has 4
lines and 20 characters per line.

Powering the WildCard

The Keypad/Display WildCard derives its power from the WildCard bus so it is automatically pow-
ered when plugged in.  It should not be hot-inserted into an active controller however.  Instead, you
should turn off power, install all WildCards, then power up your system.

Adjusting the Contrast and Attaching an External
Contrast Control

A contrast potentiometer is located on the board.  The WildCard is shipped with a high-contrast
“supertwist” LCD display that has a wide viewing angle.  In most applications the contrast/viewing
angle potentiometer can be left in its default setting, as shipped from the factory, with no adjust-
ments required.  If you wish to change the contrast setting, simply twist the potentiometer using a
small screwdriver.  While the 4x20 character displays provide good contrast over a wide viewing
angle at a fixed contrast value, the contrast may need to be readjusted if the temperature changes
significantly.

An additional three wires on the Keypad/Display cable bring out +5V, V_Contrast, and VEE/GND,
respectively.  These signals also are provided on the field header, H4.  You can connect a panel-
mounted potentiometer to these three signals on either connector if your application requires exter-
nal control of the display’s contrast and/or viewing angle.  To do so, connect the endpoints of a 10K
potentiometer to +5V and VEE/GND and run its centertap to V_Constrast.  Jumper J4 should be
removed, and J6 set to either “Char” or “Graphics” for the display type used.

Using the General Purpose I/O
Eight lines of general purpose digital I/O are provided for your use.  These lines are provided on the
WildCard’s Field Header, H4, at the pin locations shown in Table 1-4.

Table 1-4   H4: Field Header

Signal Pins Signal
GND – 1 2 – +5V
+5V – 3 4 – V+Raw

V_Contrast – 5 6 – VEE/GND
+5V – 7 8 – /BEEPER_ON

DI_7 – 9 10 – DI_6
DI_5 – 11 12 – DI_4

DIO_3 – 13 14 – DIO_2
DIO_1 – 15 16 – DIO_0

NC – 17 18 – NC
NC – 19 20 – NC
NC – 21 22 – NC
NC – 23 24 – NC

Note:
Four general purpose digital input/outputs are brought out to pins 13-
16, and four inputs to pins 9-12.

Several bytes of memory, starting at 0xC000, are used to communicate with the WildCard.  The
eight digital I/O lines are mapped to a single byte to memory address 0xC00D.  The lower nibble,
the four lines designated DIO_0 through DIO_3, are configurable as either inputs or outputs.  The
upper nibble, the four lines designated DI_4 through DI_7 are dedicated inputs.  A read of the mem-
ory address returns the inputs, a write sets the outputs.  The least significant bit, bit 0, of address
0xC00E sets the direction of the configurable nibble.  If bit 0 is set to 1 the nibble is configured as

outputs, if it is cleared to 0 the nibble is configured as inputs.  On power up the nibble defaults to
inputs.  Table 1-5 diagrams the WildCard’s memory map.

Table 1-5 Keypad/Display WildCard memory locations.

Address Bit
Positions

Meaning

0xC002 0 A write of 1 to the least significant bit turns on the beeper, 0 turns it off.

0xC002 1 A write of 1 turns on the backlight, 0 turns it off.

0xC00D 0-3 Either inputs or outputs

0xC00D 4-7 Digital inputs

0xC00E 0 A write of 1 configures the lower nibble of 0xC00D to all outputs; resetting to 0
configures the nibble to all inputs.

These locations are all read/write.  A read of an output returns the last byte written to the output, and
a read from an input returns the current state of the input pin.

When reading or writing to a single bit position you should use the kernel routines  SET.BITS and
CLEAR.BITS from Forth or compile the functions SetBits() and ClearBits() when using C.
For example, typing from a terminal connected to your controller, to enable the lower nibble of
0xC00D as outputs and to then set DIO_2 high you would first write a 1 to the lower bit of 0xC00E
as,

01 0xC00E 0x00 SET.BITS

and then set bit 2 high as,
04 0xC00D 0x00 SET.BITS

The 01 or 04 indicate the bit positions to be affected, the 0xC00E and 0xC00D indicate the 16-bit
addresses written to, and the 0x00 indicates the address page.  From your terminal you can just type
the above lines in the terminal window and send them to your controller by pressing the “Enter”
key.  The arrows on the lines above indicate the action o the “Enter” key.

The SET.BITS command takes a bit mask and address and sets the output bits corresponding to the
bits that are set in the bit mask without affecting the other output bits.  The first SET.BITS above
configures the lower nibble of the I/O port located at 0xC00D for output, and the second sets just a
single bit high.

In C you would compile the C functions SetBits() and ClearBits() into your program, com-
pile it, and download it to your controller.

Controlling the Beeper
In normal operation the beeper on the WildCard is activated in software by writing a one to the least
significant bit (LSB) at address 0xC002 on the page corresponding to the modules address (which
should normally be configured to 0).  Because other bits of the same address are used for other
functions (like the backlight), the LSB must be set and cleared using SET.BITS and CLEAR.BITS
from Forth, or using SetBits() and ClearBits() in C.  If the shorting bar is removed from

jumper J3 the beeper is disabled.  An external device capable of sinking 40 ma. can also turn on the
beeper by pulling pin 27 on header H2, or pin 8 on the field header H4, to ground.  A bipolar tran-
sistor, FET or switch can be used.

Table 1-6   H2: Keypad/Display Header

Signal Pins Signal
KPC4 – 1 2 – GND
KPC3 – 3 4 – +5V
KPR3 – 5 6 – V_Contrast
KPR0 – 7 8 – Display_A1
KPC2 – 9 10 – Display_R//W
KPR1 – 11 12 – Display_E
KPR2 – 13 14 – Display_D0
KPC1 – 15 16 – Display_D1
KPC0 – 17 18 – Display_D2

+5 – 19 20 – Display_D3
V_Contrast – 21 22 – Display_D4
VEE/GND – 23 24 – Display_D5

+5 – 25 26 – Display_D6
/BEEPER_ON – 27 28 – Display_D7

DIO_0 – 29 30 – LED Backlight+
DIO_1 – 31 32 – LED Backlight-
DIO_2 – 33 34 – VEE

Note:
Keypad connections are aligned along one side and the display
connections are aligned along the other side.  Keypad signals on odd-
numbered pins 1-17 are tapped from the cable by a 10-pin single-row
female connector (of which only 9 pins are actually used), and the
display signals on even-numbered pins 2-32 are tapped by a 16-pin
single-row female connector.

To turn on the beeper you could execute from the terminal,
01 0xC00C 0x00 SET.BITS

and to turn it off you would enter,
01 0xC00C 0x00 CLEAR.BITS

Using the Keypad
The keypad offers a simple yet effective means for an operator to control a computer-based instru-
ment.  By pushing a keypad button, the user shorts a “row” circuit to a “column” circuit.  Your con-
troller’s driver hardware and software detect the connection by reading the 4 rows as digital inputs,
and holding them in a default “high” state with pull-up resistors.  Each of the 5 columns is con-
nected to a digital output.  The processor scans each column low in turn and reads each of the row
inputs to see if it has been pulled low.  If it has, the processor deduces that a user is holding down
the key at the intersection of that row and column.

By scanning the rows and columns, the processor can identify which key was pressed.  The routines
of Table 1-7 are built into the QED-Forth kernel; they scan the keypad and report which (if any) key
is being depressed.  Using these routines we can take any desired action based on input obtained
from the keypad.  Consult the C or Forth Glossaries for detailed descriptions of these routines.

Table 1-7 Keypad driver functions.

C Name Forth Name Function

ScanKeypad ?KEYPAD Scans the keypad and if a key is being pressed it waits for a key
release, returning with the key number.  Calls PAUSE while waiting to
enable multitasking.  If no key is being pressed, it returns immediately
with a -1.

ScanKeypress ?KEYPRESS Scans the keypad returning with its current state: either –1 for no key
pressed or the key number.  Does not wait.  ScanKeypress
provides a way of monitoring the state of the keypad without “tying up”
the processor while waiting for a key to be released.  When using
ScanKeypress, make sure that multiple calls to your program do
not misinterpret a single keystroke as multiple entries from the keypad.

Keypad KEYPAD Waits for a key release and returns with the key number on key
release.  While waiting, it calls PAUSE to give other tasks a chance to
run.

These software drivers refer to keys by numbers from 0 through 19.  Key 0 is the lower right key,
Key 1 is directly above it, and Key 19 is the upper left key.  The keys are oriented as:

19 15 11 7 3

18 14 10 6 2

17 13 9 5 1

16 12 8 4 0

For example, if the user presses and releases the key in the upper right corner while the KEYPAD
routine is running, the number 3 will be returned.  For example, from the Forth prompt you can type
in,

KEYPAD .

and the controller will wait for a key press, typing the key number when it is released.

Using the Display
Your controller includes built-in software drivers for an LCD up to 4 lines by 20 characters in size.
A ribbon cable connects the board to the display.  The display is automatically initialized and
blanked upon each reset or restart, so it is ready to use at startup.

Since displays are controlled differently depending on their size and type, your controller must be
configured for the kind of display that is present.  A built-in library function named IsDisplay()
in C (or IS.DISPLAY in Forth) performs this configuration.  Fortunately, your controller stores the
display configuration information in non-volatile EEPROM that retains its data even when power is
removed, so you don’t have to reconfigure the board after each power-up.  Both character and
graphics displays are pre-configured at the factory to operate in “text mode”, so all of the sample
code described here will work without further configuration on your part.

The display interface is based on a simple idea: you write the desired characters or bit-mapped
graphics patterns to a display buffer in the controller’s RAM, and then use the pre-coded
UpdateDisplay() function (or UPDATE.DISPLAY in Forth) to transfer the contents of the buffer to
the display.   You control a character display by writing ASCII characters or strings to a 4 line by 20
character buffer in RAM using $>DISPLAY and then executing UPDATE.DISPLAY to make the con-
tents visible.

Additional pre-coded routines allow you to control the cursor and write individual data and control
symbols to the display.  To get a feel for the capabilities of these functions, you can browse through
the “Keypad/Display Interface” section of the “Categorized List of QED Library Functions” in the
“Control C Glossary” Document.  You can also take a look at the INTRFACE.H header file (in the
\FABIUS\INCLUDE\MOSAIC directory) which contains declarations for the display control func-
tions and macros.

Table 1-8 lists the display driver functions.

Table 1-8 Display driver functions.

C Name Forth Name

BufferPosition() BUFFER.POSITION

CharsPerDisplayLine() CHARS/DISPLAY.LINE

CharToDisplay() CHAR>DISPLAY and BYTES>DISPLAY

ClearDisplay() CLEAR.DISPLAY

CommandToDisplay() COMMAND>DISPLAY

DisplayBuffer() DISPLAY.BUFFER

DisplayOptions() DISPLAY.OPTIONS

DISPLAY_HEAP DISPLAY.HEAP

GARRAY_XPFA GARRAY.XPFA

InitDisplay() INIT.DISPLAY

IsDisplay() IS.DISPLAY

IsDisplayAddress() IS.DISPLAY.ADDRESS

LinesPerDisplay() LINES/DISPLAY

PutCursor() PUT.CURSOR

StringToDisplay() and STRING_TO_DISPLAY() $>DISPLAY

UpdateDisplay() UPDATE.DISPLAY

UpdateDisplayLine() UPDATE.DISPLAY.LINE

UpdateDisplayRam() (UPDATE.DISPLAY)

While there are a great many functions associated with the display, allowing you lots of flexibility in
using the display, you really need only a few to make it work.

Programming the Display in Forth

The operating system maintains an 80 character buffer whose base extended address (or xaddress) is
returned by the routine,

DISPLAY.BUFFER ( -- xaddr )

The offset from the start of this buffer to a specified line and character position is returned by the
routine,

BUFFER.POSITION   ( line#\char# -- buffer.offset )

The user can write ASCII characters into this buffer using standard operators such as C! or CMOVE, or
with the assistance of the handy utility routine

$>DISPLAY   ( x$addr\line#\char# -- )

which is pronounced “string-to-display” (in Forth, $ is often used to represent a string).

$>DISPLAY moves the string starting at x$addr to the buffer starting at the specified line number (0,
1, 2, or 3) and character position (0 through 19) in the display buffer.  This routine moves only as
many characters as will fit on the specified line.  Executing $>DISPLAY modifies the contents of the
buffer but does not alter the display.

Placing the appropriate line number (0, 1, 2, or 3) on the stack and executing

UPDATE.DISPLAY.LINE  ( line# -- )

transfers the specified line’s contents from the buffer to the display.   Executing UPDATE.DISPLAY
transfers all of the lines in the display buffer to the display.

Let’s try an example.  Type in the following definition:

: SHOW.MESSAGE ( -- )
   CLEAR.DISPLAY
   “ My favorite color is”  0 0 $>DISPLAY
   “ purple.”               1 0 $>DISPLAY
   UPDATE.DISPLAY
   ;

SHOW.MESSAGE first executes CLEAR.DISPLAY which clears the display and fills the DISPLAY.BUFFER
with ASCII blanks.  The next two lines in the definition specify the contents of the top two lines of
the display.  The characters between the quotation marks specify a string to be moved to the display
buffer, and the two numbers immediately preceding $>DISPLAY are the line number (numbered 0
through 3) and the character position (numbered 0 through 19) to which the string is moved.  If you
execute

SHOW.MESSAGE

the message will appear on the display.

To modify only the line reporting the color “purple” without changing the rest of the display, you
could execute from the terminal,

DISPLAY.BUFFER 1 0  BUFFER.POSITION XN+ 20 BLANK

“ blue.” 1 0  $>DISPLAY

1 UPDATE.DISPLAY.LINE

The first command blanks line#1 in the display, eliminating the prior contents.  BUFFER.POSITION
calculates the offset to the start of line 1, and XN+ adds this offset to the DISPLAY.BUFFER base ad-
dress to yield the start address of line# 1 in the buffer.  BLANK then blanks the 20 characters on the
line in the buffer.  The following command line moves the string “blue.” to the display buffer, and
UPDATE.DISPLAY.LINE writes the new contents of line 1 to the display.  Note that the top line of the
display is unchanged.

We could have avoided the need for the BLANK command by making the new “color” string as long
or longer than the original string on line 1.  For example, the following two commands have the
same effect as  three commands above:

“ blue.               ”  1 0 $>DISPLAY

1 UPDATE.DISPLAY.LINE

The extra trailing spaces in the string ensure that none of the prior string remains in the buffer.

To clear the display, fill the display buffer with blanks, and home the cursor to the upper left corner
of the display, simply type

CLEAR.DISPLAY

and of course executing SHOW.MESSAGE again replaces the original message on the display.

The routine INIT.DISPLAY initializes the display so that it is ready to accept characters, homes the
cursor to the upper left position, and blanks the screen, cursor, and display buffer.  This routine is
executed upon every processor reset or restart.  INIT.DISPLAY also initializes the system variable
LINES/DISPLAY to 4 and the system variable CHARS/DISPLAY.LINE to 20.  The values of these vari-
ables may be modified by the programmer to accommodate different sized displays up to a maxi-
mum of 80 characters.  For example, to interface a 2 line by 16 character display, execute

DECIMAL  2 LINES/DISPLAY !  16 CHARS/DISPLAY.LINE !

Throughout the above examples you probably noticed that no cursor was visible on the display.
This is because INIT.DISPLAY also turns the cursor off.  You have full control over the cursor, how-
ever.  The routine

DISPLAY.OPTIONS   ( display.on?\cursor.on?\cursor.blinking? -- )

sets the cursor and display state based on the three flags passed to it on the stack.  The first flag,
called display.on?, specifies whether the display is enabled or disabled.  This feature can be used
to flash the display by rapidly enabling and disabling the display via successive calls to
DISPLAY.OPTIONS.  The second flag, called cursor.on?, specifies whether the cursor is visible as
an underbar at the current cursor position.  The top flag on the stack, called cursor.blinking?,
specifies whether the cursor is visible as a blinking box obscuring the current character position.
The default condition is: display enabled, cursor off, cursor not blinking.  This condition applies
after a reset, restart, or execution of INIT.DISPLAY.  Try passing different flag combinations to
DISPLAY.OPTIONS to see what the results are.

Several lower level utilities are available to the programmer.  For example, the routine

PUT.CURSOR   ( line#\character# -- )

places the cursor at the specified location on the display.  Note that whether or not the cursor is
visible depends on the configuration set by DISPLAY.OPTIONS.  Once the cursor has been placed,
each succeeding character written to the display appears at the cursor location and causes the cursor
position to be incremented by 1.  Please note, however, that in many displays the cursor does not
advance smoothly as one might expect from the end of one line to the start of the following line!
Please test your routines carefully when using these low level utilities to control the display.

To send a single character directly to the display at the current cursor position, bypassing the display
buffer, execute

CHAR>DISPLAY   ( char -- )

This automatically increments the cursor position (but note that the cursor may skip to the start of an
unexpected line after the end of a line is reached).  CHAR>DISPLAY does not update the contents of
the DISPLAY.BUFFER.

To send a command character directly to the display, execute

COMMAND>DISPLAY   ( byte -- )

Consult the display data sheet appendix to this document to determine the numerical value associ-
ated with each valid command.  For example, the command byte that clears the display is 0x01.

The following routine uses both the keypad and display, showing on the display the key number of a
pressed key:

: TEST.KEYPAD  ( -- )
  INIT.DISPLAY    \ Initialize the display
  BEGIN
    PAUSE.ON.KEY  \ allows you to type a CR from the terminal to bail out of the loop
    “ You pressed key#:” 0 0 $>DISPLAY  \ write to buffer
    KEYPAD        \ get the key number ( -- key# )
    S>D <# # # #> DROP 1XN- ( -- x$addr ) \ convert key# to a string
    1 0 $>DISPLAY  ( -- ) \ write key# to buffer
    UPDATE.DISPLAY \ write to display
  AGAIN
  ;

After executing TEST.KEYPAD, the identifier of each key that you press is displayed as a 2-digit
number (in the current BASE) on the LCD display.  Type a carriage return at the terminal and press
one final key on the keypad to exit the routine.

The following code provides a more complete example.  It allows the user to test the display by
putting a grid of “X”s on the display, one “X” for each keypad button.  Then as the user presses each
key the corresponding “X” is removed.  It also produces a key click on each key release, showing
how to make beeps of a particular duration, and shows how to turn on the backlight.  You can
download the following code after any coldstart.

DOWNLOAD.MAP
0x04 USE.PAGE

ANEW Display.Keypad.Tester

DECIMAL \ Numbers are interpreted as decimal unless preceeded with “0x”;
        \ if preceeded with “0x” they are interpreted as hexadecimal.

\ A constant is defined for the address of the WildCard.  Page 0x00
\ corresponds to a WildCard plugged into Moduel Port 0 with no jumpers set.
\ A WildCard plugged directly into a QCard is plugged into Module Port 0.
\ Ordinarily, WildCards can be configured for module addresses 0 to 7;
\ but the Keypad/Display WildCard MUST be set to module address 0 for the
\ built-in software drivers to work.  All WildCards use 16-bit addresses
\ starting at 0x0C00.  The 16-bit value at address 0xC000 is a security key;
\ bits at address 0xC002 control the beeper (the lsb) and backlight (the next bit);
\ a bit at 0xC00E configures the I/O direction of the lower nibble of 0xC00D;
\ and 0xC00D provides 8 bits of general purpose user I/O.
\
0xC002 0x00 XCONSTANT Beeper/BacklightAddr \ WildCard MUST be located on page 0

: >Beep ( u -- | u is the number of microseconds to hold the beeper on)
   \ An audible key click is made by actuating the beeper for a short period.
   \ 1 millisecond is just enough time to give a good solid click, less time gives
   \ a weaker click.  Beeps up to 65.535 milliseconds are possible with this routine.
   0x01 Beeper/BacklightAddr SET.BITS  \ the LSB of location C002\00 controls beeper
   MICROSEC.DELAY             \ delays for u microseconds
   0x01 Beeper/BacklightAddr CLEAR.BITS \ we use SET.BITS and CLEAR.BITS so as to not
   ;                          \ affect any other bits at the address

: Show.Keypad.Test.Screen  ( -- )
   \ Shows X’s for each keypad button position.  As each button is pressed
   \ an X in the corresponding location disappears.
   “ ************ XXXXX *” 0 0 $>DISPLAY  \ Stores the string at the row\col
   “  Press keys  XXXXX *” 1 0 $>DISPLAY  \ position in the display buffer
   “   to test.   XXXXX *” 2 0 $>DISPLAY  \ UPDATE.DISPLAY then transfers the
   “ ************ XXXXX *” 3 0 $>DISPLAY  \ buffer to the screen
   UPDATE.DISPLAY
   ;

: Show.Exit.Screen   ( -- )
   \ Display screen to show after the test is done.
   “  Congratulations !  “ 0 0 $>DISPLAY  \ Stores the string at the row\col
   “                     “ 1 0 $>DISPLAY  \ position in the display buffer
   “  All the keys work  “ 2 0 $>DISPLAY  \ UPDATE.DISPLAY then transfers the
   “  as they should.    “ 3 0 $>DISPLAY  \ buffer to the screen
   UPDATE.DISPLAY
   ;

: Key>Row\Col ( key.num -- row\col )
   \ Converts a keypad button number in the range 0-19 to
   \ row and column numbers, 0-3 for rows, and 0-4 for columns.
   \ The keypad is laid out as:
   \ Row Col:   0  1  2  3  4
   \  0        19 15 11  7  3
   \  1        18 14 10  6  2
   \  2        17 13  9  5  1
   \  3        16 12  8  4  0
   4 /MOD   \ row.offset/col.offset
   3 ROT -  \ col.offset/row
   4 ROT -  \ row/col
   ;

: KEY>DISPLAY.ADDR ( key.num -- char.addr.in.display.buffer )
   Key>Row\Col 13 +  \ increment col# to start of array of X’s on the screen
   BUFFER.POSITION   \ converts row\col to a display buffer offset
   DISPLAY.BUFFER    \ leave display buffer address on stack
   ROT XN+           \ bring offset to top and add it to display buffer address
   ;

: KEYPAD.TEST ( -- )
   \ This routine waits for the user to press each key.  As each key is pressed it
   \ removes an “X” from the display at the key position.  After all the keys are
   \ pressed this routine returns. We’ll use the bit positions in a local variable to
   \ keep track of which keys are pressed, setting a different bit for each different
   \ key pressed.

   0x0000               \ for keys numbered  0-15, start with all bits cleared
   0xFFF0               \ for keys numbered 16-19, start only with four bits cleared
   LOCALS{ &msb.keys &lsb.keys } \ hold the key positions in local variables
   0x02 Beeper/BacklightAddr SET.BITS \ Turn on the backlight
   4 20 TRUE TRUE TRUE  \ Define display to have 4 rows, 20 columns, text mode,
   IS.DISPLAY           \ a char.display, and to use hitachi controller chip
   INIT.DISPLAY         \ Initializes the display, reserves the DISPLAY.BUFFER in RAM
                        \ and calls CLEAR.DISPLAY.
   Show.Keypad.Test.Screen \ Put our instructions to the user on the screen
   BEGIN
      PAUSE.ON.KEY            \ Allows premature exit with a CR from the serial port
      &msb.keys &lsb.keys XOR \ Continue until all bits are set
      WHILE
         KEYPAD            \ loops on the keypad & returns the number of pressed key
         1000 >Beep        \ makes a key click to provide audible feedback
         DUP KEY>DISPLAY.ADDR 20 -ROT C! \ place ascii blank into key’s location in
         UPDATE.DISPLAY    \ the display buffer and update the display to show it
         DUP 16 <          \ Is the key number less than 16?
         IF                \ If so, set a bit in &lsb.keys
            1 SWAP SCALE &lsb.keys OR TO &lsb.keys
         ELSE              \ If not, subtract 16 and set a bit in &msb.keys
            16 - 1 SWAP SCALE &msb.keys OR TO &msb.keys
         ENDIF
   REPEAT
   Show.Exit.Screen
   ;

Programming the Display in C

Programming the display in C is very similar to programming it in Forth.  Table 1-7 and Table 1-8
provide the corresponding function names for the keypad and display drivers.

Compiling and Downloading the HELLO.C Program

The source code for an example program discussed in the remainder of this chapter is in the
HELLO.C file in the \FABIUS\QEDCODE directory.  Here is a brief summary of how to compile the
program:

1. Open the Msoaic (TextPad) editor by double-clicking on its icon.

2. Choose “Open” from the “File” menu, and open the file named HELLO.C in the
\FABIUS\QEDCODE directory.

3. Click on the editor’s “MAKE” icon to compile the source code file and automatically generate
a download file named HELLO.TXT ready to be sent to your controller.

4. Now that the program is compiled, you can download it to the controller by entering the “Ter-
minal” program by clicking toolbar icon.  Or, if the Terminal is already active, enter it by
clicking in its window.

5. Choose “Send Text File” from the Terminal menu, change the directory to
\FABIUS\QEDCODE  by clicking on the appropriate folders, and double click on the
HELLO.TXT filename to transfer the download file to the controller.

6. You will see a hexadecimal download in Motorola S2-record format scroll across your termi-
nal window.  At the end of the download you will see some brief Forth function definitions;

these are telling the onboard operating system the names of the individual functions and vari-
ables so that we can interactively execute the functions in the HELLO.C file.

7. To run the program, simply type main from the terminal window.  Several things should hap-
pen:

When you run the program you should see the message,
Hello world!

printed in your terminal window.  You should see the following message on your LCD display:
Welcome!  Press any

keypad button to see

how the display and

keypad work together

If you press any button on the keypad, you should see the message
I’d rather be...

on the first line, followed by a message on the second line that varies depending on the column of
the keypad button that you choose.  Try it out.  To terminate the test, push a button in the left-most
column of the keypad; you’ll see the message

I’d rather be...

Done with this test.

Now if you type carriage returns from your terminal, the QED-Forth monitor will respond with the
“ok” prompt, meaning that it is ready for the next command.

You can type:
main

from your terminal any time you like, and the main function as defined in the latest download file
will be executed.

Writing to the LCD Display

Let’s take a look at the definition of the ShowMessage() function whose source code can be found
near the bottom of the HELLO.C file.  This function writes a message to the Liquid Crystal Display
(LCD).  The definition of the function is:

_Q void ShowMessage(void)
  {  STRING_TO_DISPLAY(“Welcome! Press any  “, 0, 0);
     STRING_TO_DISPLAY(“keypad button to see”, 1, 0);
     STRING_TO_DISPLAY(“how the display and “, 2, 0);
     STRING_TO_DISPLAY(“keypad work together”, 3, 0);
     UpdateDisplay();
  }

The _Q keyword declares this function as one that can be interactively called, and the void keywords
tell the compiler that this function does not expect any input parameters and does not return a value.
Each of the first four lines in the definition specifies the contents of a line on the display.  The
STRING_TO_DISPLAY() macro expects three input parameters: a string pointer, the line number,
and the character position.  It writes the specified string into the display buffer (located in the con-

troller’s RAM) starting at the specified line and character position.  Note that each of the first four
statements in the definition represents one line of the message, and the string is displayed starting at
character position 0 (the left-most position) on each line.  Lines are numbered 0 through 3 on a
character display.  Character positions are numbered 0 through 19 on a character display.

The fifth statement in ShowMessage() calls the UpdateDisplay() function which writes the
contents of the display buffer to the display.

To show the welcoming message on the display without printing a message to the terminal or run-
ning the keypad demonstration, type from your terminal:

ShowMessage( )

followed by a carriage return.  Be sure to type at least one space after the ( character.  You should
see the welcoming message appear on your LCD display.  You will also see at your terminal a line
of text that summarizes the return value of the function in several formats (decimal, hexadecimal,
and floating point), followed by the “ok” prompt.  Because the ShowMessage() function does not
return anything, the return value summary is not useful here.  The “ok” prompt indicates that con-
troller has successfully called the function and is now ready to execute another command.

For more information, consult the detailed descriptions of the STRING_TO_DISPLAY() macro and
the UpdateDisplay() function in the “Control C Glossary” in your documentation package.

Keypad

Now let’s take a look at the ManageKeypad() routine whose source code appears just before
ShowMessage() in the HELLO.C file:

_Q void ManageKeypad(void)
{  int done = 0;
   while(!done)
   {  switch((int) Keypad() / 4)
      {  case 0: idRatherBe(“Sky Diving”);   break; // col 0
         case 1: idRatherBe(“Traveling”); break; // col 1
         case 2: idRatherBe(“Watching TV”);  break; // col 2
         case 3: idRatherBe(“Eating”);    break; // col 3
         case 4: idRatherBe(“Done with this test.”);
         done = 1;            break; // col 4
      }
   }
}

As described earlier, the _Q declaration tags the function as one that will be interactively callable
from QED-Forth to speed debugging.  The void keywords tell the compiler that this function does
not return a value, and does not expect any input parameters.  The first line declares an automatic
(stack-based) variable named “done” and initializes it to zero.  The next line is a while statement
that runs until the “done” flag is true (nonzero).  The body of the while loop is a switch statement
that calls the Keypad() function which waits for a keypress and returns the index of the selected
key.  This index is divided by 4 using truncating integer arithmetic as indicated by the (int) cast to
calculate the selected keypad column number.  The column number is used by the switch statement
to select which message is displayed on the screen.  If column 4 at the left of the keypad is selected,
the “done” flag is set to true and the while statement terminates.

To interactively execute the function, simply type at your terminal:

ManageKeypad(  )

followed by a carriage return; remember to type at least one space after the ( character.  The func-
tion is now waiting for you to press a button on the keypad; once you do, you’ll see the message:

I’d rather be...

<selected string goes here>

on the display.  The routine will continue to run until you choose a key in the leftmost column of the
keypad.  Then you will see the printed summary of the routine’s return value at your terminal, fol-
lowed by QED-Forth’s “ok” prompt.  The return value summary is not meaningful in this case be-
cause the ManageKeypad() function does not return a value.



18 Keypad Display Wildcard

Keypad/Display WildCard Schematics

Figure 1-2 Keypad/Display WildCard Schematic Page 1.

Figure 1-3 Keypad/Display WildCard Schematic Page 2.

Appendix A: Display Specifications










































